Computations supporting Example 3.7 in the paper 'Gaussian quadratures with prescribed nodes via moment theory' by Rajkamal Nailwal and Aljaž Zalar.

Let m=(m_i)_{i=0,...,9} be a sequence defined by m_i=i! We will demonstrate the existence of a minimal, (2+4)-atomic representing measure, containing atoms x1=1, x2=11.

```
In[∘]:= m = Table[Factorial[k], {k, 0, 9}]
Out[∘]=
{1, 1, 2, 6, 24, 120, 720, 5040, 40 320, 362 880}
```

We will compute the localizing matrix at the polynomial f(x)=(x-1)(x-11).
locMom...localized moments at f.

```
In[∘]:= Replacements = Table[x^(9 - i) → m[[9 - i + 1]], {i, 0, 9}]
Out[∘]=
{x^9 → 362 880, x^8 → 40 320, x^7 → 5040, x^6 → 720, x^5 → 120, x^4 → 24, x^3 → 6, x^2 → 2, x → 1, 1 → 1}
```

```
In[∘]:= locMom = Table[Expand[(x - 1) (x - 11) * x^i] /. Replacements, {i, 0, 7}]
Out[∘]=
{1, -7, -26, -102, -456, -2280, -12 240, -65 520}
```

locMatrix...localized matrix H_f(3)

```
In[∘]:= locMatrix = HankelMatrix[locMom[[1 ;; 4]], locMom[[4 ;; 7]]];
        MatrixForm[locMatrix]
Out[∘]//MatrixForm=
```

$$\begin{pmatrix} 1 & -7 & -26 & -102 \\ -7 & -26 & -102 & -456 \\ -26 & -102 & -456 & -2280 \\ -102 & -456 & -2280 & -12240 \end{pmatrix}$$

We will check that H_f (3) is invertible by computing its eigenvalues.

```
In[∘]:= N[Eigenvalues[locMatrix]]
Out[∘]=
{-12 683.9, -40.5573, 3.28031, 0.136621}
```

Now we compute the polynomial g(x), i.e., the generating polynomial of the sequence locMom.
We compute the next column of H_f, restricted to given rows. Then we have to determine the kernel of the extended matrix.

In[∘]:=
```
locMatrixExt = HankelMatrix[locMom[[1 ;; 4]], locMom[[4 ;; 8]]];
MatrixForm[locMatrixExt]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & -7 & -26 & -102 & -456 \\ -7 & -26 & -102 & -456 & -2280 \\ -26 & -102 & -456 & -2280 & -12\,240 \\ -102 & -456 & -2280 & -12\,240 & -65\,520 \end{pmatrix}$$

In[∘]:=
```
v = Inverse[locMatrix].locMatrixExt[[ ;; , 5]];
u = v;
AppendTo[v, -1]
```

Out[∘]=

$$\left\{-\frac{220\,344}{1601}, \frac{1\,476\,768}{1601}, -\frac{753\,912}{1601}, \frac{95\,824}{1601}, -1\right\}$$

So the polynomial g(x) is the following.

In[∘]:=
```
g[x_] = -v.Table[x^i, {i, 0, 4}]
```

Out[∘]=

$$\frac{220\,344}{1601} - \frac{1\,476\,768\,x}{1601} + \frac{753\,912\,x^2}{1601} - \frac{95\,824\,x^3}{1601} + x^4$$

The zeroes of g are the candidates for the missing nodes.

In[∘]:=
```
N[Solve[g[x] == 0, x]]
```

Out[∘]=

$$\{\{x \to 0.162393\}, \{x \to 2.81433\}, \{x \to 5.90849\}, \{x \to 50.9674\}\}$$

Now we compute the candidate for the next moment, i.e., m_{10}.
For this aim we need to compute the product h(x) of f(x)=(x-1)(x-11) and g(x).

In[∘]:=
```
f[x_] = (x - 1) (x - 11);
h[x_] = Expand[f[x] * g[x]]
```

Out[∘]=

$$\frac{2\,423\,784}{1601} - \frac{18\,888\,576\,x}{1601} + \frac{26\,234\,592\,x^2}{1601} - \frac{11\,577\,776\,x^3}{1601} + \frac{1\,921\,411\,x^4}{1601} - \frac{115\,036\,x^5}{1601} + x^6$$

We determine m_{10} by using the L(x^4*h[x])=0, where L is the Riesz functional of the sequence m.

In[∘]:=
```
h4[x_] = Expand[x^4 * h[x]]
PrependTo[Replacements, x^(10) → t]
h4[x] /. Replacements
sol = Solve[(h4[x] /. Replacements) == 0, t]
```

Out[∘]=

$$\frac{2423784\,x^4}{1601} - \frac{18888576\,x^5}{1601} + \frac{26234592\,x^6}{1601} - \frac{11577776\,x^7}{1601} + \frac{1921411\,x^8}{1601} - \frac{115036\,x^9}{1601} + x^{10}$$

Out[∘]=

$$\{x^{10} \to t,\ x^{10} \to t,\ x^9 \to 362880,\ x^8 \to 40320,\ x^7 \to 5040,\ x^6 \to 720,\ x^5 \to 120,\ x^4 \to 24,\ x^3 \to 6,\ x^2 \to 2,\ x \to 1,\ 1 \to 1\}$$

Out[∘]=

$$-\frac{5944515264}{1601} + t$$

Out[∘]=

$$\left\{\left\{t \to \frac{5944515264}{1601}\right\}\right\}$$

In[∘]:=
```
AppendTo[m, t /. sol[[1]]]
```

Out[∘]=

$$\left\{1, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880, \frac{5944515264}{1601}\right\}$$

Let us compute M_{5}.

In[∘]:=
```
DegreeFive = HankelMatrix[m[[1 ;; 6]], m[[6 ;; 11]]];
MatrixForm[DegreeFive]
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 2 & 6 & 24 & 120 \\ 1 & 2 & 6 & 24 & 120 & 720 \\ 2 & 6 & 24 & 120 & 720 & 5040 \\ 6 & 24 & 120 & 720 & 5040 & 40320 \\ 24 & 120 & 720 & 5040 & 40320 & 362880 \\ 120 & 720 & 5040 & 40320 & 362880 & \frac{5944515264}{1601} \end{pmatrix}$$

Let us compute the eigenvalues.

In[∘]:=
```
N[Eigenvalues[DegreeFive]]
```

Out[∘]=

$$\{3.74896 \times 10^6, 5068.64, 38.0234, 1.66153, 0.349803, 0.0194753\}$$

All the eigenvalues are positive, which implies that M_5 is positive definite and the measure for m exists. Except x1 and x2, other atoms are zeroes of g(x).