
Computations supporting Example 3.6 in the paper 'Gaussian quadratures with prescribed nodes via moment theory' by Rajkamal Nailwal and Aljaž Zalar.

Let $m=(m_i)_{i=0,\dots,9}$ be a sequence defined by $m_i=i!$. We will demonstrate which conditions in the solution to the question of the existence of minimal, $(2+4)$ -atomic representing measure, containing atoms $x_1=1/3$, $x_2=11$, is violated. We first show that the measure supported on \mathbb{R} does not exist and then also if we allow an atom at infinity the answer remains the same.

```
In[4]:= m = Table[Factorial[k], {k, 0, 9}]
```

```
Out[4]= {1, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880}
```

We will compute the localizing matrix at the polynomial $f(x)=(x-1/3)(x-11)$.
locMom...localized moments at f.

```
In[5]:= Replacements = Table[x^(9 - i) → m[[9 - i + 1]], {i, 0, 9}]
```

```
Out[5]= {x^9 → 362880, x^8 → 40320, x^7 → 5040, x^6 → 720, x^5 → 120, x^4 → 24, x^3 → 6, x^2 → 2, x → 1, 1 → 1}
```

```
In[6]:= locMom = Table[Expand[(x - 1/3)(x - 11) * x^i] /. Replacements, {i, 0, 7}]
```

```
Out[6]= {-17/3, -13, -110/3, -130, -552, -2680, -14160, -75600}
```

locMatrix...localized matrix $H_f(3)$

```
In[7]:= locMatrix = HankelMatrix[locMom[[1 ;; 4]], locMom[[4 ;; 7]]];  
MatrixForm[locMatrix]
```

```
Out[8]//MatrixForm=  

$$\begin{pmatrix} -\frac{17}{3} & -13 & -\frac{110}{3} & -130 \\ -13 & -\frac{110}{3} & -130 & -552 \\ -\frac{110}{3} & -130 & -552 & -2680 \\ -130 & -552 & -2680 & -14160 \end{pmatrix}$$

```

We will check that $H_f(3)$ is invertible by computing its eigenvalues.

```
In[9]:= N[Eigenvalues[locMatrix]]
```

```
Out[9]= {-14691.9, -61.038, -1.54268, 0.176699}
```

Now we compute the polynomial $g(x)$, i.e., the generating polynomial of the sequence **locMom**.

We compute the next column of H_f , restricted to given rows. Then we have to determine the kernel of the extended matrix.

```
In[10]:= locMatrixExt = HankelMatrix[locMom[[1 ;; 4]], locMom[[4 ;; 8]]];
MatrixForm[locMatrixExt]
```

```
Out[11]//MatrixForm=
```

$$\begin{pmatrix} -\frac{17}{3} & -13 & -\frac{110}{3} & -130 & -552 \\ -13 & -\frac{110}{3} & -130 & -552 & -2680 \\ -\frac{110}{3} & -130 & -552 & -2680 & -14160 \\ -130 & -552 & -2680 & -14160 & -75600 \end{pmatrix}$$

```
In[12]:= v = Inverse[locMatrix].locMatrixExt[[ ;; , 5]];
u = v;
AppendTo[v, -1]
```

```
Out[14]=
```

$$\left\{ -\frac{46998216}{137503}, \frac{41197920}{137503}, -\frac{11282760}{137503}, \frac{1695024}{137503}, -1 \right\}$$

So the polynomial $g(x)$ is $g(x)=x^4-\frac{1695024}{137503}x^3+\frac{11282760}{137503}x^2-\frac{41197920}{137503}x+\frac{46998216}{137503}$

```
In[15]:= g[x_] = -v.Table[x^i, {i, 0, 4}]
```

```
Out[15]=
```

$$\frac{46998216}{137503} - \frac{41197920x}{137503} + \frac{11282760x^2}{137503} - \frac{1695024x^3}{137503} + x^4$$

The zeroes of g are the candidates for the missing nodes.

```
In[16]:= N[Solve[g[x] == 0, x]]
```

```
Out[16]=
```

$$\{\{x \rightarrow 1.87077\}, \{x \rightarrow 5.19637\}, \{x \rightarrow 2.63002 - 5.31442 i\}, \{x \rightarrow 2.63002 + 5.31442 i\}\}$$

However, already at this step we can see that the R-representing measure will not exist, since g has 2 complex roots and 2 reals, but it should have 4 real roots. However, let us compute the extended moment matrix $M_{\{5\}}$, since according to the theorem its positive definiteness would imply the existence of a measure. So M_5 will not be positive definite in this example.

First we compute the candidate for the next moment, i.e., $m_{\{10\}}$.

For this aim we need to compute the product $h(x)$ of $f(x)=(x-1/3)(x-11)$ and $g(x)$.

```
In[17]:= f[x_] = (x - 1 / 3) (x - 11);
h[x_] = Expand[f[x] * g[x]]
```

```
Out[18]=
```

$$\frac{172\,326\,792}{137\,503} - \frac{683\,705\,488\,x}{137\,503} + \frac{555\,278\,096\,x^2}{137\,503} - \frac{175\,284\,288\,x^3}{137\,503} + \frac{92\,991\,629\,x^4}{412\,509} - \frac{9\,760\,174\,x^5}{412\,509} + x^6$$

We determine $m_{\{10\}}$ by using the $L(x^4 \cdot h[x])=0$, where L is the Riesz functional of the sequence m .

```
In[19]:= h4[x_] = Expand[x^4 * h[x]]
PrependTo[Replacements, x^(10) -> t]
h4[x] /. Replacements
sol = Solve[(h4[x] /. Replacements) == 0, t]
```

```
Out[19]=
```

$$\frac{172\,326\,792\,x^4}{137\,503} - \frac{683\,705\,488\,x^5}{137\,503} + \frac{555\,278\,096\,x^6}{137\,503} - \frac{175\,284\,288\,x^7}{137\,503} + \frac{92\,991\,629\,x^8}{412\,509} - \frac{9\,760\,174\,x^9}{412\,509} + x^{10}$$

```
Out[20]=
```

$$\left\{ \begin{array}{l} x^{10} \rightarrow t, \quad x^9 \rightarrow 362\,880, \quad x^8 \rightarrow 40\,320, \quad x^7 \rightarrow 5040, \\ x^6 \rightarrow 720, \quad x^5 \rightarrow 120, \quad x^4 \rightarrow 24, \quad x^3 \rightarrow 6, \quad x^2 \rightarrow 2, \quad x \rightarrow 1, \quad 1 \rightarrow 1 \end{array} \right\}$$

```
Out[21]=
```

$$-\frac{492\,324\,551\,232}{137\,503} + t$$

```
Out[22]=
```

$$\left\{ \left\{ t \rightarrow \frac{492\,324\,551\,232}{137\,503} \right\} \right\}$$

```
In[23]:= AppendTo[m, t /. sol[[1]]]
```

```
Out[23]=
```

$$\left\{ 1, 1, 2, 6, 24, 120, 720, 5040, 40\,320, 362\,880, \frac{492\,324\,551\,232}{137\,503} \right\}$$

Let us compute $M_{\{5\}}$.

```
In[24]:= DegreeFive = HankelMatrix[m[[1 ;; 6]], m[[6 ;; 11]]];
MatrixForm[DegreeFive]
```

```
Out[25]//MatrixForm=
```

$$\begin{pmatrix} 1 & 1 & 2 & 6 & 24 & 120 \\ 1 & 2 & 6 & 24 & 120 & 720 \\ 2 & 6 & 24 & 120 & 720 & 5040 \\ 6 & 24 & 120 & 720 & 5040 & 40320 \\ 24 & 120 & 720 & 5040 & 40320 & 362880 \\ 120 & 720 & 5040 & 40320 & 362880 & \frac{492324551232}{137503} \end{pmatrix}$$

Let us compute the eigenvalues.

```
In[26]:= N[Eigenvalues[DegreeFive]]
```

```
Out[26]=
```

$$\{3.61774 \times 10^6, 3774.42, 14.6926, 0.808195, -0.436131, 0.0398037\}$$

There is one negative eigenvalues, which violates the condition for the existence of the measure.

Let us show that even if we allow an atom at infinity there still does not exist a 6-atomic $(R \setminus \cup\{\infty\})$ -representing measure having $x_1=1/3$ and $x_2=11$ as atoms. For this aim we need to compute the candidate for a 5-atomic R -representing measure having these two atoms for the truncation $\hat{m} = \{m_i\}_{i=0}^7$ of the given sequence m and then see whether this measure also represents m_8 and if m_9 is larger than the degree 9 moment obtained from the measure.

Recall $m = (m_i)_{i=0, \dots, 9}$ is a given sequence defined by $m_i = i!$

```
In[63]:= m = Table[Factorial[k], {k, 0, 9}]
```

```
Out[63]=
```

$$\{1, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880\}$$

\hat{m} is the degree 7 truncation of m .

```
In[64]:= hatM = Table[Factorial[k], {k, 0, 7}]
```

```
Out[64]=
```

$$\{1, 1, 2, 6, 24, 120, 720, 5040\}$$

We will compute the localizing matrix of \hat{M} at the polynomial $f(x)=(x-1/3)(x-11)$.
 locMom2 ...localized moments at f .

```
In[34]:= Replacements = Table[x^(7-i) -> m[[7-i+1]], {i, 0, 7}]
Out[34]= {x^7 -> 5040, x^6 -> 720, x^5 -> 120, x^4 -> 24, x^3 -> 6, x^2 -> 2, x -> 1, 1 -> 1}
```

```
In[36]:= locMom2 = Table[Expand[(x - 1/3) (x - 11) * x^i] /. Replacements, {i, 0, 5}]
Out[36]= {-17/3, -13, -110/3, -130, -552, -2680}
```

locMatrix2 ...localized matrix $H_f(2)$

```
In[37]:= locMatrix2 = HankelMatrix[locMom[[1 ;; 3]], locMom[[3 ;; 5]]];
MatrixForm[locMatrix2]
```

```
Out[38]//MatrixForm=
```

$$\begin{pmatrix} -\frac{17}{3} & -13 & -\frac{110}{3} \\ -13 & -\frac{110}{3} & -130 \\ -\frac{110}{3} & -130 & -552 \end{pmatrix}$$

We will check that $H_f(2)$ is invertible by computing its eigenvalues.

```
In[83]:= N[Eigenvalues[locMatrix2]]
Out[83]= {-585.517, -8.76216, -0.0537393}
```

Now we compute the polynomial $g_2(x)$, i.e., the generating polynomial of the sequence locMom2 .

We compute the next column of H_f , restricted to the given rows. Then we have to determine the kernel of the extended matrix.

```
In[65]:= locMatrixExt2 = HankelMatrix[locMom[[1 ;; 3]], locMom[[3 ;; 6]]];
MatrixForm[locMatrixExt2]
```

```
Out[66]//MatrixForm=
```

$$\begin{pmatrix} -\frac{17}{3} & -13 & -\frac{110}{3} & -130 \\ -13 & -\frac{110}{3} & -130 & -552 \\ -\frac{110}{3} & -130 & -552 & -2680 \end{pmatrix}$$

```
In[67]:= v2 = Inverse[locMatrix2].locMatrixExt2[;;, 4];
u2 = v2;
AppendTo[v2, -1]
```

```
Out[69]=
```

$$\left\{ \frac{204\,774}{1861}, -\frac{172\,062}{1861}, \frac{35\,955}{1861}, -1 \right\}$$

So the polynomial $g_2(x)$ is the following.

```
In[70]:= g2[x_] = -v2.Table[x^i, {i, 0, 3}]
```

```
Out[70]=
```

$$-\frac{204\,774}{1861} + \frac{172\,062\,x}{1861} - \frac{35\,955\,x^2}{1861} + x^3$$

The zeroes of g_2 are the candidates for the missing nodes.

```
In[71]:= N[Solve[g2[x] == 0, x]]
```

```
Out[71]=
```

$$\{\{x \rightarrow 1.81164\}, \{x \rightarrow 4.76677\}, \{x \rightarrow 12.7419\}\}$$

Let us compute the extended moment matrix $M_{\{4\}}$. According to the theorem its positive definiteness would imply the existence of a 5-atomic measure for m with x_1, x_2 as atoms.

First we compute the candidate for the next moment, i.e., $m_{\{8\}}$.

For this aim we need to compute the product $h_2(x)$ of $f(x)=(x-1/3)(x-11)$ and $g_2(x)$.

```
In[84]:= f[x_] = (x - 1 / 3) (x - 11);
h2[x_] = Expand[f[x] * g2[x]]
```

```
Out[85]=
```

$$-\frac{750\,838}{1861} + \frac{2\,951\,666\,x}{1861} - \frac{2\,286\,645\,x^2}{1861} + \frac{1\,759\,127\,x^3}{5583} - \frac{171\,139\,x^4}{5583} + x^5$$

We determine $m_{\{8\}}$ by using the $L(x^3 h_2[x])=0$, where L is the Riesz functional of the sequence hatM .

```
In[74]:= H3[x_] = Expand[x^3 * h2[x]]
PrependTo[Replacements, x^(8) -> t]
H3[x] /. Replacements
sol = Solve[(H3[x] /. Replacements) == 0, t]
```

```
Out[74]=
```

$$-\frac{750838 x^3}{1861} + \frac{2951666 x^4}{1861} - \frac{2286645 x^5}{1861} + \frac{1759127 x^6}{5583} - \frac{171139 x^7}{5583} + x^8$$

```
Out[75]=
```

$$\{x^8 \rightarrow t, x^8 \rightarrow t, x^7 \rightarrow 5040, x^6 \rightarrow 720, x^5 \rightarrow 120, x^4 \rightarrow 24, x^3 \rightarrow 6, x^2 \rightarrow 2, x \rightarrow 1, 1 \rightarrow 1\}$$

```
Out[76]=
```

$$-\frac{73385484}{1861} + t$$

```
Out[77]=
```

$$\left\{ \left\{ t \rightarrow \frac{73385484}{1861} \right\} \right\}$$

```
In[78]:= AppendTo[hatM, t /. sol[[1]]]
```

```
Out[78]=
```

$$\left\{ 1, 1, 2, 6, 24, 120, 720, 5040, \frac{73385484}{1861} \right\}$$

Let us compute $M_{\{4\}}$.

```
In[80]:= DegreeFour = HankelMatrix[hatM[[1 ;; 5]], hatM[[5 ;; 9]]];
MatrixForm[DegreeFour]
```

```
Out[81]//MatrixForm=
```

$$\begin{pmatrix} 1 & 1 & 2 & 6 & 24 \\ 1 & 2 & 6 & 24 & 120 \\ 2 & 6 & 24 & 120 & 720 \\ 6 & 24 & 120 & 720 & 5040 \\ 24 & 120 & 720 & 5040 & \frac{73385484}{1861} \end{pmatrix}$$

Let us compute the eigenvalues of M_4 .

```
In[82]:= N[Eigenvalues[DegreeFour]]
```

```
Out[82]=
```

$$\{40092.4, 86.0339, 1.69064, 0.253568, -0.0302511\}$$

There is one negative eigenvalues, which violates the condition for the existence of the measure.